# N=2 Super Lie Conformal Algebra#

The $$N=2$$ super Lie conformal algebra is an extension of the Virasoro Lie conformal algebra (with generators $$L,C$$) by an even generator $$J$$ which is primary of conformal weight $$1$$ and two odd generators $$G_1,G_2$$ which are primary of conformal weight $$3/2$$. The remaining $$\lambda$$-brackets are given by:

$\begin{split}[J_\lambda J] &= \frac{\lambda}{3} C, \\ [J_\lambda G_1] &= G_1, \\ [J_\lambda G_2] &= -G_2, \\ [{G_1}_\lambda G_1] &= [{G_2}_\lambda G_2 ] = 0, \\ [{G_1}_\lambda G_2] &= L + \frac{1}{2} TJ + \lambda J + \frac{\lambda^2}{6}C.\end{split}$

AUTHORS:

• Reimundo Heluani (2020-06-03): Initial implementation.

class sage.algebras.lie_conformal_algebras.n2_lie_conformal_algebra.N2LieConformalAlgebra(R)#

The N=2 super Lie conformal algebra.

INPUT:

• R – a commutative ring; the base ring of this super Lie conformal algebra.

EXAMPLES:

sage: F.<x> = NumberField(x^2 -2)
sage: R = lie_conformal_algebras.N2(F); R
The N=2 super Lie conformal algebra over Number Field in x with defining polynomial x^2 - 2
sage: R.inject_variables()
Defining L, J, G1, G2, C
sage: G1.bracket(G2)
{0: L + 1/2*TJ, 1: J, 2: 1/3*C}
sage: G2.bracket(G1)
{0: L - 1/2*TJ, 1: -J, 2: 1/3*C}
sage: G1.degree()
3/2
sage: J.degree()
1


The topological twist is a Virasoro vector with central charge 0:

sage: L2 = L - 1/2*J.T()
sage: L2.bracket(L2) == {0: L2.T(), 1: 2*L2}
True


The sum of the fermions is a generator of the Neveu-Schwarz Lie conformal algebra:

sage: G = (G1 + G2)
sage: G.bracket(G)
{0: 2*L, 2: 2/3*C}